
Применение таймера для формирования временных интервалов
Задача генерирования прямоугольных импульсов сводится к формированию временных интервалов и изменению логических уровней на выводах МК в конце каждого временного интервала. Наиболее гибкое решение получается с использованием прерываний по переполнению таймера. Тогда команды, изменяющие логические уровни на выво​дах МК, помещаются внутри процедуры обработки прерываний. В разных случаях может быть удобно формирование интервала, равного периоду, а чаще половине периода. Тогда частота повторения процедуры вдвое превышает генерируемую частоту.
Длительность интервала, по истечении которого будет выработано прерывание, определяется тактовой частотой контроллера и содержимым таймера в начальный момент времени. Чем больше начальное значение таймера, тем быстрее произойдет его перепол​нение и тем меньше длительность интервала. Если в начальный момент времени таймер содержал нулевое значение, то прерывание произойдет через 65536·0,2·10 6с ~ 13,1 мс, где 0,2·10– 6с — длительность машинного цикла. Предполагается, что тактовая частота равна 20 МГц и предварительный делитель таймера отсутствует. Изменяя начальное значение таймера можно сформировать временной интервал любой длительности с абсолютной по​грешностью теоретически не более ±100 не.
Рассмотрим пример использования прерываний. Программа, приведенная ниже, обеспечивает генерирование на всех линиях порта B прямоугольных импульсов частотой примерно 985 Гц. Для формирования временных интервалов используется прерывание от таймера 1.

;Описание РСФ
status           equ             03       ;Регистр состояния
porta             equ             05       ;Порт A
portb             equ             06       ; Порт B
intcon           equ             0B       ; Регистр управления прерываниями
pir1               equ             0C       ; Регистр флагов прерываний
pie1               equ             8C       ; Регистр управления периферийными прерываниями
tmr1l             equ             0E       ; Младший байт таймера 1
tmr1h             equ             0F       ; Старший байт таймера 1
t1con             equ             10       ; Регистр управления таймером 1
; Описание РОН
w_temp         equ               20       ; Регистры для временного. . .
stat_temp equ               21       ; ...хранения состояния
1.       goto             start
2.       org               4                            ; Процедуры обработки прерываний
3.       movwf           w_temp        ;Сохранение состояния
4.       swapf           status, w
5.       movwf           stat_temp
6.       comf             portb, f      ;Инверсия порта B
7.       bcf               t1con, 0      ; Остановить таймер 1
8.       movlw           0F6
9.       movwf           tmr1h
10.      movlw           016
11.      movwf           tmr1l
12.      bsf               t1con, 0      ; Остановить таймер 1
13.      bcf               pir1, 0       ; Сброс флага прерывания
14.      swapf           stat_temp, w  ;Восстановление состояния
15.      movwf           status
16.      movf             w_temp, w
17.      retfie
start                                                         ; Основная программа
18.       movlw           b'00111111'   ; Настройка на ввод...
19.       tris             porta        ; ...порта A
20.       movlw           b'00000000'   ; Настройка на вывод...

21.       tris             portb        ;•••порта B
22.       clrf             portb        ;Сброс в ноль линий порта B
23.       movlw           b'00000001'   ; Настройка и запуск таймера 1
24.       movwf           t1con
; Настройка прерываний
25.       movlw           b'11000000'  ; Разрешение периферийных прерываний
26.       movwf           intcon
27.       bsf               status, 5     ; Выбор страницы памяти 1
28.       movlw           b'00000001'   ; Разрешение прерывания от таймера 1
29.       movwf           pie1
30.       bcf               status, 5     ; Возврат на страницу памяти 0
loop
31.       goto             loop          ; Вечный цикл
end
Первая команда необходима для того, чтобы «обойти» процедуру обработки прерываний и приступить к выполнению основной программы. На начало основной програм​мы указывает метка START (команда №18). Команды №18-22 подготавливают порты к работе. Все линии порта A настраиваются на ввод путем записи единиц в регистр TRISA командой TRIS; порт B настраивается на вывод.
Команды №23, 24 настраивают таймер 1. Здесь используется двоичное представление числа. Такая запись упрощает задание каждого бита РСФ. 
Настройка прерываний (команды №25-30) выполняется в два этапа. Во-первых, устанавливаются общий бит разрешения прерываний и бит разрешения периферийных прерываний в регистре INTCON. Во-вторых, разрешается прерывание от таймера 1 через регистр PIE1. Следует учитывать, что PIE1 расположен на первой странице памяти, поэтому прежде чем обратиться к нему необходимо переключить стра​ницу (команда №27), а после обращения выбрать исходную нулевую страницу (команда №30).
Далее контроллер переходит к выполнению вечного цикла. Команда №31 выпол​няет переход на саму себя. Цикл будет периодически приостанавливаться прерыванием от таймера 1.
Рассмотрим процедуру обработки прерываний (команды №3-17). На второй стро​ке расположена директива ассемблера ORG, задающая адрес следующей команды. Проце​дура начинается с сохранения состояния (команды №3-5). Эти команды включены в про​грамму лишь для примера. Здесь сохранять состояние не обязательно, потому что основ​ная часть программы представляет собой вечный цикл и не использует ни аккумулятора, ни регистра STATUS.
Изменение уровней на выводах порта B выполняется командой COMF (логиче​ская инверсия). Таким образом, каждый раз при переполнении таймера уровни порта B меняются на противоположные. Очевидно, что интервал времени от одного переполнения до другого соответствует половине периода генерируемых импульсов.
Команды №7-15 служат для присваивания таймеру начального значения, опреде​ляющего частоту выходных импульсов. Рассчитаем начальное значение таймера.
Желаемая частота f=985 Гц, период T=0,001015228 c. Определим количество N машин​ных циклов контроллера, соответствующих половине периода генерируемого сигнала.

где 0,2·10–6 — длительность машинного цикла микроконтроллера при тактовой частоте 20 МГц. Начальное значение таймера определяется как разность модуля счета таймера M и числа N. Обозначим начальное значение таймера буквой C, тогда C=M-N=216-2538 = 62998. Это число необходимо разместить в двух регистрах TMR1H и TMR1L, по​этому переходим к шестнадцатеричной записи числа: 6299810=F61616 и выделяем старший и младший байты: F6 и 16.
Такой расчет является неточным, потому что не учитывает время, затрачиваемое на выполнение команд №3-12. Проводить точный расчет нецелесообразно. Для этого пришлось бы вручную пересчитать все команды с начала процедуры обработки прерываний до запуска таймера, учитывая при этом время их выполнения (1 или 2 машинных цикла). На практике начальное значение таймера рассчитывается приближенно, период повтора определяется экспериментально при помощи отладочных средств, а затем корректируется (подгоняется).
Присваивание таймеру начального значения выполняется в два этапа. Сначала старший байт (команды №8, 9), затем младший (команды №10, 11), хотя можно и наоборот. В этой связи следует отметить важную особенность. Запись в таймер ведется побайт​но, поэтому в какой-то момент таймер содержит некоторое промежуточное значение — старший байт обновлен, а младший еще нет. Если в этот момент очередной инкремент таймера вызовет перенос из младшего байта в старший, то результирующее значение тай​мера будет отличаться от желаемого. Проиллюстрируем такую ситуацию примером. Пусть требуется записать в таймер шестнадцатеричное число 102016. В таблице приведен фрагмент программы и значения таймера в каждом машинном цикле. Для удобства машинные циклы пронумерованы.
Таблица 1 
	№ Маш. цикла
	Команда
	Содержимое таймера
	Примечание

	
	
	TMR1H
	TMR1L
	

	1
	movlw    10
	00
	FE
	Таймер содержит 00FE

	2
	movwf    tmr1h
	10
	FF
	Старшему байту присвоено значе​ние 10

	3
	movlw    20
	11
	00
	Младший байт переполняется, вы​зывая инкремент старшего байта

	4
	movwf    tmr1l
	11
	20
	Таймер содержит 1120 вместо 1020


Чтобы предотвратить такую ошибку нужно останавливать таймер, прежде чем
обращаться к нему для чтения или записи, а затем снова запускать. В рассматриваемом
примере программы для этого служат команды №7, 12. Существует и другой способ —
очищать командой CLRF младший байт таймера перед записью старшего байта. Но тогда
необходимо соблюдать очередность записи: сначала старший байт, затем младший.
Команда №13 очищает флаг прерывания, предотвращая его повторную обработку. Команды №14-16 восстанавливают состояние контроллера. Выход из процедуры обработки прерываний осуществляется командой RETFIE.

